
Abstract 

FI is a program written in 1980 for the HP-41C programmable calculator to evaluate the definite integral between given limits of an 

arbitrary user-supplied function f(x) using the 3-point Gauss-Legendre quadrature formula applied over a number of subintervals. 

Three worked examples are included. 

1. Introduction 

FI is a short (56 steps) RPN program that I wrote in 1980 for the HP-41C programmable calculator (will also run 

as-is in the HP-41CV/CX and the HP42S), which can evaluate the definite integral between specified limits of an 

arbitrary user-supplied function f(x) using the fast 3-point Gauss-Legendre quadrature formula applied over a 

given number of subintervals. The method is as follows: we want to compute 

𝐼 =    f(x)
b

a

.𝑑𝑥 

but first of all the change of variable 𝑥 = 1

2
 b + a + 1

2
 b − a t ,  dx = 1

2
 b − a .𝑑𝑡 transforms the interval (a, b) into 

the interval (-1, 1). The 3-point Gauss-Legendre quadrature formula then gives: 

which is exact for polynomial f(x) up to the 5
th

 degree, and a 5
th

-order approximation otherwise. See Notes. 

 

 

2. Program Listing 

 

01 ♦LBL “FI”   15  RCL 01   29  RCL 03  43  XEQ IND 00 - 55 steps 

02 “NAME?”   16   /    30   +   44   8   - requires at least SIZE 006  

03  AON    17  STO 05   31  XEQ IND 00 45   *        - uses the Alpha register 

04  PROMPT   18   2    32   5   46  ST- 02  - uses no flags and doesn’t alter  

05  AOFF    19   /    33   *   47  RCL 05        angular mode or display settings 

06  ASTO 00   20  ST+ 04   34  ST- 02  48  ST+ 04  - to get  *  press the [x] key 

07 “N?”    21   .6    35  RCL 04  49  DSE 01     - to get  /  press the [] key 

08  PROMPT   22  SQRT   36  RCL 03   50  GTO 00 ►       - the symbols ♦ and ► are purely 

09  STO 01   23   *    37   -   51  RCL 02   cosmetic, to indicate branching 

10 “a↑b?”     24  STO 03   38  XEQ IND 00 52   *     

11  PROMPT     25  CLX    39   5   53   18 

12 ♦LBL “FIP”  26  STO 02   40   *   54   / 

13  STO 04   27 ♦LBL 00   41  ST- 02  55  END 

14   -      28  RCL 04   42  RCL 04  

3. Usage Instructions 

The program can be used both interactively and programmatically, as follows: 

 

1) Interactively:  in RUN mode, call “FI” (Function Integrator). The program will prompt for the name 

of the program which defines f(x), the required number of subintervals (n) and the integration limits (a, 

b). Once provided, the program will proceed to compute the integral and return its value to the display. 

  

2) Programmatically:  your program must call “FIP” (Function Integrator Programmable), which 

assumes that the name of the program which defines f(x) is stored in register R00 , the number of 

subintervals (n, which should be an integer ≥ 1) is stored in register R01 , the lower limit (a) is in stack 

register Y and the upper limit (b) is in stack register X.  
 

Upon returning control to your program, the computed value of the integral will be in stack register X. 

The name of your f(x) will remain unaltered in register R00 but the number of subintervals will be lost. 

No subroutine levels are used apart from calls to your f(x). 

 

In both cases you need to write a program to define f(x), the equation to solve. It must be an independent 

program under its own global label, must assume that the argument x is in stack register X upon being called, 

and must return the computed value of f(x) in stack register X as well.  

 

4. Examples 

The following examples can be useful to check that the program is correctly entered and to understand its usage. 

 

4.1 Example 1 

Evaluate the integral 𝐼 =    x520.19

3.59
.𝑑𝑥 

 

In PRGM Mode, enter the following 4-step program to define the function f(x) to be integrated: 

 

01 ♦LBL “FX1”  02  5   03  Y↑X   04  END 

 

In RUN Mode, proceed as follows to evaluate the integral using just one (sub)interval: 

 

XEQ “FI”   NAME? “FX1”  [R/S]  N?   1  [R/S]  a↑b?   3.59 [ENTER] 20.19 [R/S]   11288934.08 

 

The theoretical value is (20.19
6 

- 3.59
6
)/6 = 11288934.08... so the computed value has all 10 digits correct. The 

reason for such accuracy despite the large integration interval and using just one (sub)interval is because the 

3-point Gauss-Legendre quadrature formula is exact for polynomial functions up to the 5
th

 degree, which x
5
 is. 

 

4.2 Example 2 

Evaluate the integral  𝐼 =    
Sin (x)

x

1

0
.𝑑𝑥 

 

In PRGM Mode, enter the following 5-step program to define the function f(x) to be integrated: 

 

01 ♦LBL “FX2”  02  SIN     03  LASTX      04  /    05  END 

 

In RUN Mode (and RAD, FIX 9), proceed as follows to evaluate the integral using 1, 2 and 4 subinterval(s): 

XEQ “FI”   NAME? “FX2”  [R/S]  N?   1  [R/S]  a↑b?   0 [ENTER] 1 [R/S]   0.946083134  (339) 

XEQ “FI”   NAME? “FX2”  [R/S]  N?   2  [R/S]  a↑b?   0 [ENTER] 1 [R/S]   0.946083072  (717) 

XEQ “FI”   NAME? “FX2”  [R/S]  N?   4  [R/S]  a↑b?   0 [ENTER] 1 [R/S]   0.946083071  (706) 

 

The exact value rounds to 0.9460830704 so the computed values have 6, 8 and ~10 correct digits, respectively. 

 

4.3 Example 3 

 

Write a program to compute the volume of the solid of revolution generated by any user-specified curve and in 

particular use it to find the volume of the solid of revolution obtained by the turn of the catenary 

 𝑦 =  
3𝑒𝑥/3  + 3𝑒−𝑥/3

2
   

around the x axis between x = 0 and x = 1.2  (your program will call “FIP” (Programmable) to do the hard work.) 
 

 

In PRGM Mode, enter the following 20-step generic program to compute said volume for any curve y=f(x): 

 

01 ♦LBL “VOL”    05  AOFF  09   2  13  XEQ “FIP”  17 ♦LBL “VAUX”  

02 “NAME?”    06  ASTO 06 10  STO 01 14   PI   18  XEQ IND 06 

03  AON     07 “VAUX” 11 “X1↑X2?”   15   *   19  X↑2 

04  PROMPT    08  ASTO 00 12  PROMPT  16  RTN   20  END 

 

Now, still in PRGM Mode, enter the following 10-step program to define the catenary’s formula: 

 

01 ♦LBL “CATEN” 03   /  05  ENTER↑  07   +    09   *    

02   3     04  E↑X  06  1/X   08  1.5  10  END 

 

In RUN Mode, proceed as follows to evaluate the volume asked for: 

 

XEQ “VOL”  NAME? “CATEN”  [R/S]   X1↑X2?    0 [ENTER] 1.2 [R/S]   35.79755410  (correct to ~9 places) 

       

 

Notes 

1. Despite its very simple coefficients, this 3-point Gauss-Legendre quadrature formula gives 5th-order accuracy for arbitrary 

f(x) using just 3 evaluations per subinterval. This is much better than Simpson’s Rule, which only gives 3rd-order accuracy. 

2. Integrals which have singularities at one or both endpoints can also be computed, as f(x) is not evaluated there. 

3. If you can use synthetic instructions, you may replace registers R03, R04 and R05 by registers M, N, and O respectively (i.e.: 

STO 03 becomes STO M and so on), and insert step 55 CLA just before END to clear the Alpha register before the program ends. 

After this, program length will be 56 steps and min. SIZE 003, thus saving 3 numbered registers for other uses. 

4. This program (FI, FIP) was duly submitted for inclusion in the PPC ROM but it wasn’t accepted. 

